

```

10 REM      MOP - MANAGEMENT ORDERING AND PRODUCTION SYNTHESIS
20 REM      COPYRIGHT (C) 1980
30 REM
40 REM      SPECIFICATION BY
50 REM      B H CLAYBOURNE
55 REM      SCHOOL OF OPERATION MANAGMT.
60 REM
65 REM      PROGRAM BY
70 REM      R REED
71 REM      CBL ADVISOR
75 REM      COMPUTER UNIT
76 REM
77 REM
80 REM      THE POLYTECHNIC
85 REM      NEWCASTLE UPON TYNE
90 REM
98 POKE 144,49
99 REM NEW SESSION: INITIAL CONDITIONS, VARIABLES AND CONSTANTS
100 DIM S(33),R(36),A(33),L(21),T(33)
200 M(1)=400:M(2)=60:M(3)=40:M(4)=60:M(5)=150
205 Y$(1)="      DAY":Y$(2)=" EVENING":Y$(3)="SATURDAY":Y$(4)="  SUNDAY"
208 Y$(5)="SUBCONTR"
210 C(1)=4:C(2)=4.3:C(3)=4.4:C(4)=5:C(5)=4.4
220 E(1)=.5:E(2)=1:E(3)=2:E(4)=3:E(5)=3.5
230 R(1)=2.5:I(1)=.02:G(1)=.04:L(20)=1
240 F(1)=0.5:F(2)=1:F(3)=1.5:F(4)=2:F(5)=3:F(6)=4.5:F(7)=6:F(8)=8
250 FOR K= 13 TO 32:READ S(K):NEXT K
260 DATA 320,420,400,380,420,380,520,500,440,400
270 DATA 490,500,260,530,490,540,530,325,340,530
290 A(1)=.2:A(2)=.1:A(3)=1:A(4)=.3:A(5)=.05

```

```

310 P$(3)="STRIKE:100% PROD. LOST":S=0.5
320 Q(1)=.25:Q(2)=.15:Q(3)=.05
350 S$(1)="50% RAW MATERIAL DELIVERY, THE REST      NEXT WEEK"
360 S$(2)="RAW MATERIALS DELAYED 1 WEEK"
370 S$(3)="3 WEEK CYCLE FOR DELIVERY"
390 S$(5)="2 WEEK CYCLE FOR DELIVERY"
394 REM
395 REM  NEW GAME :INITIAL CONDITIONS
397 GOSUB 8920:REM PRINTER ATTACHED?
400 PRINT"*** M O P S ***"
410 PRINT"PRODUCTION TEAMS NAME ";
415 GOSUB 9800
416 N$=Z$
420 G(2)=200:I(2)=550:A(12)=340:S(12)=340:R(4)=0:R(2)=0
430 M(7)=325:M(6)=325:R(13)=800:N=2:W=12
440 FOR K=1 TO 19:L(K)=0:NEXT K
450 FOR K=13 TO 33:A(K)=0:NEXT K
460 FOR K=14 TO 33:R(K)=0:NEXT K
470 FOR K=1 TO 33:T(K)=0:NEXT K
994 REM
995 REM  NEW WEEK :INITIAL CONDITIONS
1000 W=W+1:IF W<=32 THEN 1015
1010 PRINT"END OF THIS GAME":GOTO 6510
1013 REM
1015 R(4)=R(2):R(2)=0
1016 B=RND(-TI):REM STARTS A NEW RANDOM NUMBER SEED
1018 E(7)=0:C(7)=0:R(3)=0:I(3)=0:G(3)=0:F(9)=0:C(8)=0:M(6)=0
1020 FOR K=1 TO 5:P(K)=0:NEXT K
1023 IF W<16 THEN 1305
1024 REM
1025 REM  SUPPLY ALTERATIONS
1030 S$(7)=" "
1040 B=INT(100*RND(2))
1050 IF B>7 OR L(1)>1 THEN 1060
1054 IF R(W) THEN S$(7)=S$(1):L(1)=L(1)+1
1056 R(W)=R(W)*S:R(W+1)=R(W+1)+R(W)*(1-S)
1058 GOTO 1070
1060 IF B>17 OR L(2)>1 THEN 1070
1064 IF R(W) THEN S$(7)=S$(2):L(2)=L(2)+1
1066 R(W+1)=R(W+1)+R(W):R(W)=0
1070 B=RND(3):IF B>.07 OR L(3)>1 THEN 1090
1074 L(3)=L(3)+1:N=3
1090 IF W>25 THEN N=2
1294 REM
1295 REM  COMMAND REQUEST
1300 REM
1305 PRINT:PRINT:PRINT"COMMAND(OR PRESS H FOR HELP)";
1306 REM
1307 REM  COMMAND DECODER
1310 GOSUB 9800:PRINT:PRINT
1320 IF Z$="H" THEN 2010
1325 IF Z$="R" THEN 2510
1330 IF Z$="D" THEN X=1:P(X)=0:GOTO2610
1335 IF Z$="E" THEN X=2:P(X)=0:GOTO2610
1340 IF Z$="S" THEN X=5:P(X)=0:GOTO2610
1345 IF Z$="A" THEN X=3:P(X)=0:GOTO2610
1350 IF Z$="U" THEN X=4:P(X)=0:GOTO2610
1355 IF Z$="X" THEN 400
1360 IF Z$="C" THEN 1400
1365 IF Z$="Z" THEN 6510
1390 PRINT:PRINT"COMMAND NOT RECOGNISED":GOTO 1305
1400 PRINT"*** CURRENT VALUES ***":PRINT
1410 FOR K=1 TO 5:PRINT:PRINT Y$(K);" SHIFT",P(K):NEXT K
1430 PRINT:PRINT"MATERIAL ORDER",R(2)
1440 PRINT:PRINT:PRINT:PRINT "TO SUBMIT THE PLAN TYPE C"
1450 PRINT:PRINT"TO CHANGE THE PLAN TYPE H":PRINT

```

```

1460 GOSUB 9800
1470 IF Z#="C" THEN 3020
1480 IF Z#="H" THEN 2010
1490 GOTO 1460
2000 REM
2005 REM      HELP
2010 PRINT"J      ***** HELP *****"
2011 PRINT"M      THE FOLLOWING LIST OF COMMAND KEYS"
2012 PRINT"HAVE THE GIVEN MEANINGS.NO OTHER KEYS"
2013 PRINT"WILL WORK."
2014 PRINT"MH - DISPLAYS THE COMMAND KEYS"
2015 PRINT"MR - SIGNALS A RAW MATERIAL ORDER"
2016 PRINT"MD - SIGNALS PRODUCTION FOR DAY SHIFT"
2017 PRINT"E - SIGNALS PRODUCTION FOR EVE. SHIFT"
2019 PRINT"A - SIGNALS PRODUCTION FOR SAT. SHIFT"
2020 PRINT"U - SIGNALS PRODUCTION FOR SUN. SHIFT"
2021 PRINT"S - SIGNALS PRODUCTION FROM SUBCONTRACT"
2022 PRINT"MC - SIGNALS A PRODUCTION SUBMISSION"
2050 GOTO 1305
2495 REM
2497 REM      VARIABLES INPUT
2500 REM RAW MATERIAL ORDER
2510 PRINT"RAW MATERIAL ORDER IS ";
2520 GOSUB 9800
2530 R(2)=VAL(Z#): IF R(2)>=0 AND R(2)<=3000 THEN 2550
2540 PRINT:PRINT"NUMBER OUTSIDE LIMITS OF 0 AND 3000"
2550 GOTO 1305
2600 REM
2605 REM      PRODUCTION REQUIREMENTS
2610 GOSUB 9000
2615 IF L(21)=1 THEN 2630
2620 PRINT Y$(X);
2625 GOSUB 8020
2630 GOTO 1305
3000 REM
3010 REM AVAILABILITY OF SUBCONTRACTING
3020 IF P(5)=0 THEN L(13)=0:GOTO 3137
3030 IF L(13)=1 AND L(14)=1 THEN 3060
3040 IF L(13)=1 THEN 3070
3050 B=RND(8): IF B<0.4 THEN L(13)=1:P(5)=0:GOTO 3137
3060 L(13)=0:GOTO 3137
3070 B=RND(2): IF B<0.4 THEN L(14)=1:P(5)=0
3133 REM
3135 REM PRODUCTION AND QUALITY CONTROL
3137 P=0:Q=0
3138 IF W>24 AND W<27 THEN 3260
3140 IF W<14 THEN 3270
3150 B=RND(2):IF B>=.5 THEN 3245
3155 REM STRIKE CONDITIONS
3160 IF L(10)=1 AND L(17)=1 THEN 3210
3170 IF L(10)=1 THEN 3200
3180 IF B<.06 THEN P=A(3):L(10)=1:GOTO 3260
3190 GOTO 3210
3200 L(17)=1:B=RND(7):IF B<.5 THEN P=A(3):GOTO 3260
3205 REM PRODUCTION LLOSS FACTORS
3210 IF B<=.09 AND L(8)<4 THEN P=A(1):L(8)=L(8)+1:GOTO 3260
3220 IF B<=.24 AND L(9)<4 THEN P=A(2):L(9)=L(9)+1:GOTO 3260
3230 IF B<=.30 AND L(11)<4 THEN P=A(4):L(11)=L(11)+1:GOTO 3260
3235 IF B<=.49 AND L(15)<4 THEN P=A(5):L(15)=L(15)+1:GOTO 3260
3240 GOTO 3260
3243 REM QUALITY LOSS FACTORS
3245 IF B<=.6 AND L(5)<4 THEN Q=Q(1):L(5)=L(5)+1:GOTO 3260
3250 IF B<=.7 AND L(6)<4 THEN Q=Q(2):L(6)=L(6)+1:GOTO 3260
3255 IF B<=.99 AND L(7)<4 THEN Q=Q(3):L(7)=L(7)+1:GOTO 3260
3258 REM
3259 REM QUANTITY      CALCULATIONS

```



```

3260 IF P=1 THEN Q=0
3265 REM MAKE ADJUSTMENTS
3270 FOR K=1 TO 4:P(K)=INT(P(K)*(1-P(Q))):NEXT K
3280 M(7)=P(1)+P(2)+P(3)+P(4)+P(5)
3290 R(W+N)=R(W+N)+R(2)
3300 IF L(10)=1 OR L(10)=2 OR W=25 OR W=26 THEN C(7)=600 :GOTO 3320
3310 FOR K=2 TO 5:C(7)=C(7)+P(K)*C(K):NEXT K
3315 C(7)=C(7)+1600
3320 IF P(5) THEN C(7)=C(7)+100
3330 IF Q THEN I(2)=I(2)+R(W)-M(6):GOTO 3340
3335 I(2)=I(2)+R(W)-M(7)
3340 G(4)=G(2)
3350 G(2)=G(2)+M(7)
3360 IF S(W)<=G(2) THEN G(2)=G(2)-S(W):A(W)=S(W):F=0:GOTO 3455
3370 A(W)=G(2):G(2)=0:F=S(W)-A(W)
3373 REM
3375 REM SHORTAGE PENALTIES
3380 IF F<11 THEN F(9)=INT(F*F(1)+0.9)*4.45:GOTO 3455
3390 FOR K=1 TO 6
3400 IF F<=K*50 THEN F(9)=INT(F*F(K+1)+0.9)*4.45 :GOTO 3455
3410 NEXT K
3420 F(9)=INT(F*F(8)+.9)*4.45
3445 REM
3450 REM UNDER PROD. PENALTIES
3455 M(8)=M(7)-P(5) W=25 or 26 THEN 3510
3460 IF M(8)>=396 THEN 3510
3470 IF M(8)>=349 THEN E(7)=(400-M(8))*E(1):GOTO 3510
3480 IF M(8)>=299 THEN E(7)=(400-M(8))*E(2):GOTO 3510
3490 IF M(8)>=249 THEN E(7)=(400-M(8))*E(3):GOTO 3510
3495 IF M(8)>=199 THEN E(7)=(400-M(8))*E(4):GOTO 3510
3500 E(7)=(400-M(8))*E(4)
3505 REM
3506 REM CALCULATION TOTALS
3510 G(3)=G(2)*G(1)
3514 R(3)=R(W)*R(1)
3516 I(3)=I(2)*I(1)
3525 C(8)=INT(600+1.05*A(W)+.9)
3530 T(1)=C(7)+R(3)+I(3)+G(3)+C(8)+E(7)
3540 T(W)=A(W)*13.5
3545 IF R(2) THEN T(1)=T(1)+30
3550 T(5)=T(5)+S(W):T(6)=T(6)+A(W):T(7)=.01*INT(100*(T(7)+T(1)))
3560 T(8)=T(8)+F(9):T(9)=T(9)+T(W):T(10)=.01*INT(100*(T(9)-T(7)))
3570 T(11)=T(11)+C(8)
4000 REM
4010 REM
4015 PRINT"J"
4020 PRINT" WEEKLY REPORT WEEK";W
4030 PRINT"MIN THE WEEK YOU HAVE JUST SUBMITTED"
4040 PRINT"THE FOLLOWING OCCURRED"
4050 GOSUB 7820
4060 IF P THEN PRINT"PRODUCTIVITY FACTOR ";(1-P)*100;"%"
4070 IF Q THEN PRINT"QUALITY LOSS FACTOR ";(1-Q)*100;"%"
4080 IF L(10)=1 THEN PRINT P$(3):P$(3)=" "
4090 IF L(10)=1 AND L(17)=1 THEN PRINT P$(4):P$(4)=" "
4100 PRINT"PRODUCTION QUANTITIES:"
4110 FOR K=1 TO 4:IF P(K) THEN PRINT,,Y$(K),P(K):NEXT K
4120 IF L(13) AND P(5)=0 THEN PRINT"SUBCONTRACTING FAILED TO DELIVER"
4130 IF P(5) THEN PRINT ,Y$(5),P(5)
4135 PRINT
4140 PRINT" TOTAL PRODUCTION ";M(7)
4150 PRINT"GOODS INVENTORY BEFORE DEMAND";G(4)+M(7)
4155 GOSUB 7820
4158 PRINT
4160 PRINT"RAW MATERIAL RECEIVED ";R(W)
4170 IF N=3 THEN PRINT S$(3):GOTO 4185
4180 PRINT S$(5)

```

```

4185 PRINT S$(7)
4190 PRINT "SCHEDULED DELIVERY NEXT WEEK ";R(W+1)
4195 GOSUB 7820
4197 PRINT
4198 PRINT
4200 PRINT"DEMAND QUANTITY WAS ",S(W)
4210 PRINT"DEMAND SATISFIED WAS ",R(W)
4220 PRINT"FINISHED GOODS STOCK ",G(2)
4230 PRINT"RAW MATERIALS STOCK ",I(2)
4240 IF F THEN PRINT"CUSTOMER SHORTAGES INCURRED WERE ";F
4260 PRINT:PRINT:PRINT"RECORD THIS DATA"
4270 GOSUB 7820
5000 REM
5010 REM LAST WEEKS FIGURES
5020 PRINT"J      LAST WEEKS FIGURES"
5030 PRINT"MTYPE IN YOUR CALCULATIONS WHEN REQUESTED"
5035 IF R(2) THEN PRINT "RAW MATERIAL ORDER COST  = 30"
5040 IF R(W) THEN Y$="MATERIAL RECIEVED COSTS  =" :Y=R(3):GOSUB 8520
5050 IF I(2) THEN Y$="RAW MATERIAL HOLDING COSTS=" :Y=I(3):GOSUB 8520
5060 IF G(2) THEN Y$="PRODUCT HOLDING COSTS    =" :Y=G(3):GOSUB8520
5070 IF F THEN Y$="SHORTAGE COSTS              =" :Y=F(9):GOSUB 8520
5080 REM INPUT VALUES
5083 IF W=25 OR W=26 THEN PRINT" HOLIDAY SHUTDOWN COSTS = 600":GOTO5150
5085 IF(W<25ORW>26)AND(L(10)=0ORL(10)=3)THENPRINT:PRINT" DAY SHIFT COST  =1600"
5090 FOR K=2 TO 4
5100 IF P(K)=0 THEN 5140
5110 Y$=Y$(K)+" SHIFT PROD COST="
5120 Y=P(K)*C(K)
5130 GOSUB 8520
5140 NEXT K
5150 IF P(5)=0 THEN 5200
5160 Y$=Y$(5)+" PROD COST (ADD 100)="
5170 Y=P(5)*C(5)+100
5190 GOSUB 8520
5195 IF E(7) THEN PRINT:Y$="UNDER PRODUCTION PENALTY=" :Y=E(7):GOSUB8520
5200 IF C(8) THEN PRINT:Y$="      INDIRECT COSTS  =" :Y=C(8):GOSUB 8520
5210 IF P=1 THEN PRINT"STRIKE COSTS = 600"
5215 PRINT
5220 PRINT"TOTAL COSTS          =",T(1)
5230 PRINT"TOTAL CASH SALES   =",T(W)
5240 Y$="PROFIT =" :Y=T(W)-T(1):GOSUB 8523
5260 PRINT:PRINT "CUMULATIVE : COSTS      =",T(7)
5270 PRINT:PRINT "          PROFITS   =",T(10)
5280 PRINT:PRINT "          SHORTAGE  =",T(8)
5700 GOSUB 7820
5710 GOTO 1000
6500 REM
6505 REM FINAL ANALYSIS
6510 IF L(20)=1 THEN 6530
6515 IF W=13 THEN 6720
6518 OPEN 4,4
6520 CMD 4
6530 PRINT"      CUMULATIVE ANALYSIS  WEEK";W;"■"
6540 PRINT"TOTAL : "
6560 PRINT"      SALES MET          ",T(6)
6570 PRINT"      SALES DEMAND       ",T(5)
6580 PRINT"CUSTOMER SHORTAGES    ",T(5)-T(6)
6590 PRINT"CUMULATIVE PROD COSTS ",T(7)
6600 PRINT"CUMULATIVE IND. COSTS  ",T(11)
6610 PRINT"TOTAL SALES REVENUE      ",T(9)
6620 PRINT"PROFIT                  ",T(10)
6630 PRINT"TOTAL SHORTAGE COSTS    ",T(8)
6640 PRINT"OPTIMUM PROFIT          ",INT(T(6)+.9)*4.45
6645 IF T(6)=0 THEN PRINT:GOTO 6670
6650 PRINT"PROFIT FACTOR          ",INT(100*(T(10)/(T(6)*4.45))/100
6660 IF T(8)THENPRINT"SURVIVAL FACTOR",INT(100*(T(6)*4.45)/T(8))/100:GOTO 6670

```

```

6665 PRINT"SURVIVAL FACTOR---FANTASTIC"
6670 PRINT
6675 IF L(20)=1 THEN 6720
6700 PRINT#4
6710 CLOSE 4
6720 GOTO 9990
7800 REM
7810 REM WAIT
7820 PRINT:PRINT:PRINT"PRESS THE SPACE BAR TO CONTINUE"
7830 GET Z$:IF Z$(">")="" THEN 7830
7835 PRINT
7840 RETURN
8000 REM
8010 REM PRODUCTION INPUT SUB.
8020 PRINT" PRODUCTION REQUIRED =";
8030 GOSUB 9800
8035 IF Z$="" THEN 8040
8035 IF VAL(Z$)=0 THEN PRINT:PRINT"INPUT A NUMBER":GOTO 8020
8040 P(X)=VAL(Z$):IF P(X)>=0 AND P(X)<=M(X) THEN 8050
8045 PRINT:PRINT"NUMBER OUTSIDE LIMITS OF 0 AND";M(X):P(X)=0:GOTO 8070
8050 M(6)=M(6)+P(X):IF M(6)<=I(2) THEN 8070
8060 PRINT"TOTAL PRODUCTION OF ";M(6);"IS GREATER THAN THE RAW MATERIAL"
8061 PRINT"INVENTORY OF ";I(2)
8065 PRINT" CHANGE YOUR PRODUCTION REQUESTS":FOR K=1 TO 5:P(K)=0:NEXT K:M(6)=0
8070 RETURN
8500 REM
8510 REM SUB TO CHECK RESPONSES
8520 Y=INT(Y*100)/100
8525 L(12)=0
8525 PRINT:PRINT Y$;
8530 GOSUB 9800
8540 L(12)=L(12)+1
8550 IF ABS(VAL(Z$)-Y)<.01 THEN PRINT:PRINT"CORRECT":GOTO 8590
8560 IF L(12)<2 THEN PRINT:PRINT"INCORRECT, TRY AGAIN":PRINT:GOTO 8525
8563 IF L(20) THEN PRINT:PRINT "CORRECT VALUE=";Y:GOTO 8595
8565 OPEN 4,4
8568 PRINT#4,
8570 PRINT#4,N$,"WEEK";W
8575 PRINT#4,Y$
8580 PRINT#4,"YOUR ANSWER";VAL(Z$),"CORRECT ANSWER";Y
8585 PRINT#4,
8590 CLOSE4
8595 PRINT
8600 RETURN
8900 REM
8910 REM CHECK FOR PRINTER ATTACHED
8920 PRINT"J"
8925 PRINT" IS THE PRINTER ATTACHED AND SWITCHED":PRINT"ON ?"
8930 GOSUB 9800
8940 IF LEFT$(Z$,1)="Y" THEN L(20)=0:GOTO 8970
8950 IF LEFT$(Z$,1)="N" THEN L(20)=1:GOTO 8970
8960 PRINT"PLEASE ANSWER YES OR NO":GOTO 8925
8970 RETURN
9000 REM
9010 REM TEST FOR HOLIDAY PERIOD
9020 L(21)=0:IF (W<25 OR W>26) OR X=5 THEN 9050
9030 PRINT:PRINT"HOLIDAYS: ONLY SUBCONTRACTING ALLOWED":L(21)=1
9050 RETURN
9780 REM
9790 REM INPUT SUB.
9800 OPEN 1,0,1:INPUT#1,Z$:CLOSE1:IF Z$="" THEN 9800
9810 RETURN
9990 POKE 144,46
9993 NEW
9995 END

```

READY.